To install click the Add extension button. That's it.

The source code for the WIKI 2 extension is being checked by specialists of the Mozilla Foundation, Google, and Apple. You could also do it yourself at any point in time.

4,5
Kelly Slayton
Congratulations on this excellent venture… what a great idea!
Alexander Grigorievskiy
I use WIKI 2 every day and almost forgot how the original Wikipedia looks like.
Live Statistics
English Articles
Improved in 24 Hours
What we do. Every page goes through several hundred of perfecting techniques; in live mode. Quite the same Wikipedia. Just better.
.
Leo
Newton
Brights
Milds

# File:Gamma distribution pdf.svg

## Summary

 Description The probability density function of Gamma distribution for different parameters. See below for the spec and Gnuplot source code. Date 26 June 2010, 18:31 (UTC) Source Gamma_distribution_pdf.png Author Gamma_distribution_pdf.png: MarkSweep and Cburnett derivative work: Autopilot (talk)

## Spec and Gnuplot Source Code

For the readers who are interested in Erlang distribution, we have prepared another specific image (see right). Note that in Erlang distribution the shape parameter k must take integers.

Originally, the pdf of Gamma distribution is

${\displaystyle f(x;k,\theta )={\frac {x^{k-1}e^{-{\frac {x}{\theta }}}}{\theta ^{k}\Gamma (k)}}.}$

Let α = k, and β = 1/θ. Then

${\displaystyle f(x;k,\theta )=f(x;\alpha ,1/\beta )={\frac {\beta ^{\alpha }x^{\alpha -1}e^{-\beta x}}{\Gamma (\alpha )}}.}$

Hence

${\displaystyle \ln f(x;k,\theta )=\alpha \ln \beta +(\alpha -1)\ln x-\ln \Gamma (\alpha )-\beta x}$

which directly gives the code below.

 This W3C-unspecified plot was created with Gnuplot.
# Author: Autopilot, Ika
# Date: 12/08/2014

set term svg size 800,600 font 'Times, 24' enhanced
set output 'gamma_dist_pdf.svg'
set samples 1001

set xtics 0,2
set ytics 0,0.1
set mxtics 10
set mytics 10
set yrange [0:0.5]
set key top right Left reverse

# Ln f(x; k, theta)
# Commonly, we denote a = alpha = k and b = beta = 1/theta.
ln_f(x, a, b) = a*log(b) - lgamma(a) + (a-1)*log(x) - b*x

# f(x; k, theta)
f(x, k, theta) = (x<0)? 0 : (x==0)? ((k<1)? 1/0 : (k==1)? theta : 0)\
: (theta==0)? 0 : exp(ln_f(x, k, 1.0/theta))

plot [0:20] \
f(x, 1.0, 2.0) t "k = 1.0, {/Symbol q} = 2.0" lw 2 lc rgb "red", \
f(x, 2.0, 2.0) t "k = 2.0, {/Symbol q} = 2.0" lw 2 lc rgb "orange", \
f(x, 3.0, 2.0) t "k = 3.0, {/Symbol q} = 2.0" lw 2 lc rgb "yellow", \
f(x, 5.0, 1.0) t "k = 5.0, {/Symbol q} = 1.0" lw 2 lc rgb "green", \
f(x, 9.0, 0.5) t "k = 9.0, {/Symbol q} = 0.5" lw 2 lc rgb "black", \
f(x, 7.5, 1.0) t "k = 7.5, {/Symbol q} = 1.0" lw 2 lc rgb "blue", \
f(x, 0.5, 1.0) t "k = 0.5, {/Symbol q} = 1.0" lw 2 lc rgb "purple"

set term x11
set output


 This is a retouched picture, which means that it has been digitally altered from its original version. Modifications: Recreated as svg with gnuplot. The original can be viewed here: Gamma distribution pdf.png. Modifications made by Autopilot.
I, the copyright holder of this work, hereby publish it under the following licenses:
 Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.2 or any later version published by the Free Software Foundation; with no Invariant Sections, no Front-Cover Texts, and no Back-Cover Texts. A copy of the license is included in the section entitled GNU Free Documentation License.
 This work is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or any later version. This work is distributed in the hope that it will be useful, but without any warranty; without even the implied warranty of merchantability or fitness for a particular purpose. See version 2 and version 3 of the GNU General Public License for more details.